

POSTERS

E-Mail OS/390 SAS/MXG Computer Performance Reports in HTML Format

Neal Musitano Jr.
Department of Veterans Affairs
Information Technology Center
Philadelphia, Pennsylvania

ABSTRACT

This paper displays a user example to generate and to e-mail an OS/390 SAS® report in HTML format. The report is created and e-mailed via a batch job using BASE SAS®. Input to this example is from the daily OS/390 SAS/MXG Performance Database, but e-mail reports can be generated from your choice of input data.

The example e-mail report is generated as an HTML file using the Version 8.2 BASE SAS® Output Display System (ODS) and PROC REPORT. Reports can also be generated via other procedures such as PROC PRINT.

INTRODUCTION

The SAS/MXG software processes the OS/390 SMF/RMF performance data into daily, weekly and monthly MXG® Computer Performance Databases.

These databases contain numerous SAS® datasets. Each SAS® dataset contains performance variables in a SAS® readable format. The advantage of using MXG® software is that it converts raw performance data from IBM and third party computer vendors into SAS® readable format for processing by SAS® jobs.

MXG® provides numerous examples of SAS® analyze examples and plots, from utilizing SAS/BASE to using SAS/GRAPH®.

PREPARATION

Figure #1 displays the JCL used for this example. Two e-mail options were used. The option "EMAILSYS=SMTP" specifies the SMTP e-mail interface and 'EMAILHOST=123.456.789.2' specifies the TCP/IP address of the mainframe e-mail server. The option 'EMAILPORT=25' to identify the default e-mail port number, might be needed at some sites.

These options worked for this example. However; depending on your base SAS procedure setup and system TCP/IP installation you may want to check with your computer site if specifying the parameters is necessary or to verify your computer site's specific parameter information.

Figure #1 OS/390 SAS JCL

```

//*****
//S1 EXEC SASV82,
// OPTIONS='EMAILSYS=SMTP EMAILHOST=123.456.789.2'
//***
//LIBRARY DD DSN=MXG.FORMATS,DISP=SHR
//SOURCLIB DD DSN=MXG.SOURCLIB,DISP=SHR
//PDB1 DD DSN=MXG.PDB.DAILY(-1),DISP=SHR
//PDB2 DD DSN=MXG.PDB.DAILY(0),DISP=SHR
//***
//*** USE SAS TO SEND E-MAIL
//*** E-MAIL REPORT
//***
//SYSIN DD DSN=MXG.USER.SOURCLIB(PMXGMAIL),
// DISP=SHR

```

The SAS web site is often a good place to begin a search for information. The following link <http://support.sas.com/admin/mainframe/os390/tips/index.html> provided an excellent reference for examples and tips on OS/390 e-mail.

POSTERS

Figure #2 E-Mail Report in HTML Format - From OS/390 Base SAS®.




Figure #2 displays an e-mail report generated from BASE/SAS® Version 8.2 using the Output Delivery System (ODS). The code to produce the e-mail report is displayed in Figure#3A-F. A modified test MXG® PDB was used for input data, no actual values of detailed computer performance data were used in the paper. .

The HTML e-mail display is being viewed from Microsoft Outlook. Note that some internet service providers, have firewalls that prevent HTML e-mails or have their own e-mail set up that might perform unexpected edits on HTML e-mail.

The e-mail report displays the OS/390 CPU usage for each shift for two days of input. The date variable with the label “date of starttime” is used as a break variable.

The “TO”, “CC”, “REPLYTO”, and “SUBJECT” are specified in the SAS program that writes the TEXT/HTML file. Note that for this example the from-id is the “user” of the batch job, but a different “REPLYTO” id can be specified.

The min, mean and max values of CPU usage are displayed for each shift of each date displayed.

The OS/390 value for the computer sysplex-id is taken from the input data and is displayed in the second title line by using #BYVALS. The footnote displays the OS/390 job name that generated the report and the date that the job started execution.

POSTERS

E-MAIL Program Details.

The SAS code for the report is displayed in Figures #3A-F and used ODS and PROC REPORT to generate the e-mail report. The SAS code is one program but is sectioned into Figures #3A-F for display purposes.

The NOBYLINE option is used, since the program will use #BYVALS in the TITLE statement to print the BY values. Optionally PROC TEMPLATE was used to edit the default style to change the background color of the HTML report from a gray background to a light blue color CXCCFFFF.

Figure #3A

```
/* USE SAS TO SEND E-MAIL      */
/* MEMBER PMXGMAIL             */

OPTIONS NOBYLINE ;

/* change HTML background color */
PROC TEMPLATE;
edit styles.default as styles.testA1;
  style color_list /
    'bgA' = CXCCFFFF;
  end;
RUN;
```

The daily MXG® pdb.rmfintrv data was used for input to the program. Interval computer performance data is summarized in this dataset.

The DATA step creates a “MYRMFDATA” SAS dataset and uses the SET statement to reference two OS/390 DD cards in the JCL of Figure#1. Each PDB1 and PDB2 file contains the SAS/MXG dataset RMFINTRV. Since two days of input to the data step were used, the date variable was used as a break variable in the report.

PROC SORT is used with the NODUP option, as a precautionary double check to prevent duplicate data being used in the report. The input data “MYRMFDATA” is then sorted by sysplex, date, and shift for the report.

Figure #3B

```
/* input two days data into report */
/* select sysplex                  */
/* sort by sysplex date shift      */

DATA MYRMFDATA;
  SET PDB1.RMFINTRV
      PDB2.RMFINTRV
      (where=(sysplex='SUGPLEX') );
RUN;

PROC SORT NODUP DATA=MYRMFDATA;
  BY SYSPLEX DATE SHIFT;
RUN;
```

POSTERS

PROC MEANS is used to find the min, mean and maximum values of the variable “PLATBUSY” for each daily shift. The noprint option was used to prevent undesired printing of output from the means procedure. Output is directed to a work SAS® dataset named “MYEMAILDATA.”

Figure #3C

```
/* find min mean max platbusy      */
/* for each daily shift            */

PROC MEANS NOPRINT DATA=MYRMFDATA;
          BY SYSPLEX DATE SHIFT;
VAR PLATBUSY ;
  OUTPUT OUT=MYEMAILDATA
  MIN= MINPLATBUSY
  MEAN= AVGPLATBUSY
  MAX= MAXPLATBUSY ;
RUN;
```

The FILENAME statement defines “MYFILE” will be written as e-mail.

Additional items defining “TO”, “CC”, “REPLYTO”, and “SUBJECT” are specified and the file type is defined as a “TEXT/HTML” file. Note, that multiple e-mail id’s can be specified in the “TO” and “CC” fields.

Please use your e-mail id in the “REPLYTO” address. Note: the REPLYTO e-mail address is not necessarily the FROM e-mail address.

The Output Display System (ODS) is creating a HTML file with a body name MYFILE. Note “MYFILE” is referenced in the previous filename statement as an e-mail type file.

The RS=none option is used for OS/390 systems. It suppresses a record separator and causes a new record for each HTML output line. However; you might consider testing an example without specifying it. The STYLE of TESTA1 is being used – note it was created from the initial PROC TEMPLATE edit of the default style.

Figure #3D

```
/* Define the e-mail file          */

FILENAME MYFILE EMAIL
  TO= ("PersonAEmailid@VA.GOV"
       "PersonBEMAILid@VA.GOV")
  CC= (
       "PersonCEMAILid@VA.GOV")
  REPLYTO= ("YOUREMAILID@ABCD.COM")
  SUBJECT="SAS MXG E-MAIL REPORT"
  TYPE="TEXT/HTML" ;
RUN;

ODS HTML BODY=MYFILE RS=NONE STYLE=TESTA1;
```

POSTERS

This example uses PROC REPORT to generate the report, since it provides numerous options for formatting the output and can be easily expanded on for future report modifications. Input to the procedure is the "MYEMAILDATA" file from the previous PROC MEANS. Note that PROC PRINT could also be used as quick and easy way to produce the output.

The title statements use colors. The text color and background colors are specified on both title statements. Since a "BYVAL1" is used, the sysplex id (the first by variable in the PROC and the sort selection) is displayed on the second title line in Figure #2.

The column statement creates a column for the variables that are used in the report. The define statements defines for each variable detailed output control and format options.

Figure #3E

```
/* USE PROC REPORT TO REPORT      */
PROC REPORT SPLIT='*' DATA=MYEMAILDATA
  HEADLINE HEADSKIP
  STYLE(REPORT)=
    {BACKGROUND=RED borderwidth=3
    bordercolor=blue cellspacing=2}
  STYLE(HEADER)=
    {FOREGROUND=MAROON BACKGROUND=CXFFFFFF};
  BY SYSPLEX;

LABEL
  MINPLATBUSY='MIN PLATFORM*PERCENT*HARDWARE BUSY'
  AVGPLATBUSY='AVG PLATFORM*PERCENT*HARDWARE BUSY'
  MAXPLATBUSY='MAX PLATFORM*PERCENT*HARDWARE BUSY';

TITLE1 COLOR=BLUE BCOLOR=CX00FF00
  'CPU USAGE REPORT FROM SAS/MXG SOFTWARE';

TITLE2 COLOR=WHITE BCOLOR=ORANGE
  ' SYSPLEX = #BYVAL1';

COLUMN
  DATE SHIFT MINPLATBUSY AVGPLATBUSY MAXPLATBUSY;

DEFINE DATE / ORDER ORDER=DATA
  FORMAT=DATE9. STYLE={FOREGROUND=BLUE
  BACKGROUND=GOLD FONT_WEIGHT=BOLD}
  CENTER;
DEFINE SHIFT /
  DISPLAY FORMAT=$CHAR1. CENTER
  STYLE={FOREGROUND=BLACK BACKGROUND=ORANGE
  FONT_WEIGHT=BOLD };

DEFINE MINPLATBUSY / DISPLAY FORMAT=COMMA9.0 CENTER;

DEFINE AVGPLATBUSY / DISPLAY
  FORMAT=COMMA9.0 CENTER
  STYLE={FOREGROUND=BLACK BACKGROUND=CX99FF99
  FONT_WEIGHT=BOLD};

DEFINE MAXPLATBUSY / DISPLAY FORMAT=COMMA9.0 CENTER;
```

POSTERS

A footnote statement is used to place a footnote at the bottom left of the report. The SAS® jobid and date the job started execution are displayed in the footnote.

Figure #3F

```
FOOTNOTE1 COLOR=BLUE
          JUSTIFY=LEFT HEIGHT=2.4
"REPORT FROM JOB = &SYSJOBID ON &SYSDATE9";

/* NOTE THE DOUBLE QUOTES ON FOOTNOTE1 */

RUN;
ODS HTML CLOSE;
```

CONCLUSION

The OS/390 Base SAS JCL procedure is easily used to send SAS® based e-mail reports. The daily, weekly and monthly reports are e-mailed to specific individuals on time and improved the ease of use for managers and technicians to review computer performance activity.

Providing e-mailed BASE SAS® HTML reports from PROC REPORT or PROC PRINT also improved management and customer satisfaction by eliminating misrouted printed reports.

REFERENCES

IBM OS/390 - MVS System Management Facilities (SMF) - GC28-1783-05
IBM OS/390 - Resource Measurement Facility (RMF) User's Guide - SC28-1949-03.
Merrill's Expanded Guide to Computer Performance Evaluation Using the SAS System.
Technical Newsletters for Users of MXG®.
MXG Archives at WWW.MXG.COM
NESUG Proceeding 1994.
Effective Poster Presentations for SAS® User Group Conferences: Helen-Jean Talbott
SAS Technical Support at <http://support.sas.com>
SAS-L archives at <http://www.listserv.uga.edu/>
SAS ONLINE DOC Version 8 CDROM.

CONTACT INFORMATION

Your comments and questions are welcome.

Neal Musitano Jr.
Department of Veterans Affairs
Information Technology Center-284/31
P. O. BOX 7545
Philadelphia, PA. 19101-7545

Phone 215-842-2000 ext. 4102
FAX 215-381-3456
EMAIL TSDNMUSI@VBA.VA.GOV

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are trademarks of their respective companies.